

Tumor Assessment using RECIST: Behind the Scenes

Suhas R. Sanjee, Accenture LLP, Philadelphia, PA
Rajavel Ganesan, Accenture LLP, Philadelphia, PA
Mary N. Varughese, Accenture LLP, Philadelphia, PA

ABSTRACT

Tumor assessment is a vital part of drug discovery in oncology franchise. Most of the efficacy endpoints such as Progression Free Survival, and Time to Progression are based on tumor assessments. Tumor assessments can be quantitative and qualitative. Various steps involved in quantitative tumor assessments are discussed in this paper. In quantitative tumor assessment, patient scans/images are fed into an algorithm, which quantifies the tumor based on its characteristics. A number of image processing algorithms are available for tumor quantification from Computed Tomography (CT), Positron Emission Tomography (PET), Magnetic Resonance Imaging (MRI) and other modalities. These algorithms are traditionally not implemented in SAS® and the statistical programmer will have these assessments pre-calculated in SDTM/raw data for analysis. This paper describes the implementation of a simple tumor quantification and assessment algorithm using SAS procedures. Once the tumor is quantified, tumor assessments are derived based on the RECIST criteria. Implementing an end-to-end tumor assessment algorithm in-house using SAS provides the capability to perform sophisticated exploratory analysis of assessments derived from patient scans. Another advantage is that the pharmaceutical / medical device companies can submit the SAS code used for end-to-end tumor assessment to the regulatory agencies and thus enhance transparency. This is especially beneficial when primary and key secondary endpoints of the trial are based on tumor assessments.

INTRODUCTION

Tumor assessments can be quantitative and qualitative. Qualitative tumor assessments involve a radiologist looking at the patient scans and providing an assessment based on his expertise and experience. Qualitative tumor assessment is not discussed in this paper.

In quantitative tumor assessment, the tumor is identified and measured using automated/semi-automated algorithms. Semi-automated algorithms are more popular than fully automated ones since the latter are not error free. Patient scans are performed over time and the tumor measurement at different time-points are used to derive overall patient assessment based on RECIST criteria. Tumor assessments can be performed from images acquired using different modalities such as CT, PET, MRI, and etc.

A semi-automated algorithm for tumor assessments is presented in the following sections.

DATA OVERVIEW

The patient scans used for this exercise were downloaded from Public Lung Database to Assess Drug Response housed @ < <http://www.via.cornell.edu/databases/crpf.html>>. The images are in DICOM format and are read into SAS for analysis (Figure 1).

CT scan of chest is used below as an illustration to segment a lung nodule. All lung nodules are not cancerous. In order to identify the lung nodule, the radiologist identifies the cancerous nodule and provides the seed point.

SAS PROGRAMMING OVERVIEW

This section will discuss the steps involved in implementing proposed algorithm using SAS. The IML procedure in conjunction with DATA step is used for tumor quantification and to derive the overall patient assessment.

READING DICOM IMAGES

DICOM stands for Digital Imaging and Communication in Medicine. DICOM image data (pixel intensities) is read into SAS using INFILE statement.

The code snippet to read pixel intensities is shown in Figure 1. Similar logic can be applied to read other information such as patient details, image height, image width and etc. The DICOM header varies from one manufacturer to the other. The code snippet in Figure 1 might have to be tweaked for images from different manufacturers.

```

FILENAME img2 "C:\1.2.826.0.1.3680043.2.656.4.1.1.896.55.dcm";
DATA image;
  KEEP tag y l dicm ;
  INFILE img2 RECFM=N lrecl=32700;
  IF _N_=1 THEN DO;
    DO i=1 TO 16;
      INPUT prelude pib8.;
    END;
    INPUT dicm $4.;
  end;
  INPUT x1 $1. x2 $1. x3 $1. x4 $1.;
  tag='('||PUT(x2||x1,hex4.)||','||PUT(x4||x3,hex4.)||')';
  IF tag="(7FE0,0010)" THEN DO;
    INPUT l pib4.;
    DO i=1 to l;
      INPUT y ib2.;
      OUTPUT;
    END;
  END;
  ELSE input +(-3);
RUN;

```

Figure 1: SAS Code to Read data from DICOM pixel intensities

ALGORITHM FOR TUMOR QUANTIFICATION

Various steps involved in tumor detection and quantification are described in this section.

The algorithm for tumor quantification consists of 3 steps:

1. Image Enhancement
2. Segmentation of Lung Nodule
3. Quantification

Each of these steps is described in detail below.

1. Image Enhancement

Pixel values of CT images are the CT numbers which represent X-ray attenuation co-efficient of the bones/tissues that are imaged. Pixel values of CT images have a wide range - [-2000 to +2000] for the images used here (Figure 3). Visualizing the entire range of CT numbers in an image presents a problem as displays typically show about 250 shades of gray of which less than 100 are visually discernible [1].

Water is the reference material for CT number and it has a value of 0. Tissues or materials with attenuation (density) greater than water will have positive CT numbers. Those that are less dense will have negative CT numbers.

In order to focus on the tissues of interest and for better visualization, windowing and contrast stretching are usually employed in CT image analysis and visualization.

Windowing

The window is the range of CT numbers that will be displayed with the different shades of gray, ranging from black to white. Tissues within the window will have different shades of gray (brightness) and will have visible contrast. All tissues and materials (such as water surrounding the tissue etc.) that have CT numbers above the window will be all white (32767) and all that have CT numbers below the window will be all black (0). A window of length 2000 and level of -500 is used here to better visualize lungs and associated structures.

Contrast Stretching

The resulting pixels after windowing will range from -500 to 1500. The pixel intensities for a 16 bit image ranges from 0 to 32767. In order to better visualize, the pixel values are transformed from a range of [-500 – 1500] to [0 – 32767] using contrast stretching. Windowing and contrast stretching are implemented in PROC IML as illustrated in Figure 2.

```

*Implement Windowing and Contrast Stretching;
PROC IML;
  USE ctpixels;
  *Read input image from SAS data set into matrix;
  READ all var {pixel} into temp;
  *Reshape the matrix;
  im=shape(temp,&nrow,&ncol);
  im1=im;

  *Enhance the lung nodule portion of CT;
  DO i=1 to &nrow;
    DO j=1 TO &ncol;
      IF im[i,j]<=-500 THEN im1[i,j]=0;
      ELSE if im[i,j]>=1500 THEN im1[i,j]=32767;
      im1[i,j]=round(((im[i,j]+500)/max(im))*32767),1.0);
    END;
  END;
QUIT;

```

Figure 2: SAS Code to implement Windowing and Contrast Stretching



Figure 3: CT Image of Lungs before Windowing and Contrast Stretching



Figure 4: CT Image of Lungs after Windowing and Contrast Stretching

It can be noticed from that Figure 4 provides much better visualization (contrast) than Figure 3 and emphasizes the lung nodule/tumor.

2. Segmentation of Lung Nodule:

The lung nodule is segmented using region growing algorithm. The seed point is identified by the radiologist. Starting from the seed point, neighboring pixels having same or similar intensities are identified programmatically using contour tracing or region growing algorithms[2]. Implementation of a simple region growing algorithm in SAS is illustrated below in Figure 5.

Region growing algorithm uses eight-point connectivity in conjunction with thresholding criteria to segment the nodule. Eight pixels that are immediate neighbors of each pixel starting from the seed point are checked for thresholding criteria. If any of them satisfies thresholding criteria, they are added to the region. The loop for region growing algorithm is terminated if previous 100 pixels did not meet the thresholding and connectivity criteria.

Macro variables INITCOL and INITROW contain X and Y co-ordinates of the seed point. THRESHVAL is assigned to 25% of the seed point's pixel intensity after image enhancement. NROW & NCOL contain number of rows and columns in the input image respectively.

```

PROC IML;
  USE dicom;*Read image data after enhancement;
  READ all var {pixel} into temp;
  im=shape(temp,&nrow,&ncol); im1=im; *Reshape the matrix;
  region=j(&nrow,&ncol,0); *Define the matrix to store region;
  queue={&initrow &initcol};final={&initrow &initcol 0 0 0 0 0 0 0 0 0 0};
  regval=im[&initrow,&initcol];
  sum=0; nrow=1;
  *Region Growing Algorithm;
  DO loop=1 to %EVAL(&nrow*&ncol) while ((loop<=100 | sum>100));
    *the first queue position determines the new values;
    xv = queue[1,1];
    yv = queue[1,2];
    IF nrow(queue)>1 THEN
      queue=queue[2:nrow(queue),1:2];
    nque=nrow(queue);
    *Check neighboring points from the current seed point
    and check if they satisfy threshold for the region;
    DO i=-1 TO 1;
      DO j=-1 TO 1;
        cpx=xv+i; cpy=yv+j;
        ubound=regVal - &thresVal;
        lbound=regVal + &thresVal;
        cim=im[xv+i,yv+j];
        cregval=region[xv+i, yv+j];
        dist=sqrt((xv+i-&initrow)*(xv+i-&initrow) + (yv+j-&initcol)*(yv+j-
          &initcol));
        IF (((xv+i)>0) & ((xv+i)<=&nRow)) & /*within the x-bounds*/
          (yv+j>0 & yv+j<=&nCol) & /*within the y-bounds*/
          (i^=0 | j^=0) & /*i/j of (0/0) is redundant*/
          (region[xv+i, yv+j] ^= 1) & /*pixelposition already set?*/
          dist < &maxDist & /*within distance*/
          im[xv+i,yv+j]<=(regVal + &thresVal) & /*within range*/
          im[xv+i,yv+j]>=(regVal - &thresVal) then do;/*of the threshold*/
          /*current pixel is true, if all properties are fulfilled*/
          region[xv+i, yv+j] = 1;
          true=1;
          /*add the current pixel to the computation queue (recursive)*/
          cpx=xv+i; cpy=yv+j; nm=j(1,2,0);
          nm=cpx[1] || cpy[1];
          IF nrow(queue)>=1 then DO;
            queue = queue // nm;
          END;
          ELSE queue={cpx cpy};
        END;
      ELSE true=0;
      nmf=cpx[1] || cpy[1] || ubound[1] || lbound[1] || cim[1] ||
        regVal[1] || cregval || i[1] || j[1] || true[1] || dist[1] ||
        nque[1]; final=final // nmf;
    END;
  END;
  nrow=nrow // nque[1];
  IF loop>=100 THEN
    sum=sum(nrow[loop-100+1:loop]);
  end;
  /* Create SAS Data set */
  CREATE region FROM final; APPEND from final; CLOSE region;
QUIT;

```

Figure 5: SAS Code to implement Region Growing Algorithm

Data set region contains the X co-ordinates (COL1), Y co-ordinates (COL2) of image pixels and a flag (COL10) to indicate whether the pixel is part of lung nodule or not.




Figure 6: CT Image of Lungs Showing the seed point identified by the radiologist




Figure 7: Segmented Nodule

Figure 6 shows the CT image slice with the seed point (in red) identified by the radiologist. Figure 7 shows the segmented lung nodule. You can see that there are some dark spots or "holes" in Figure 7 due to the pixels in those patches not meeting the threshold criteria of region growing algorithm. However, this doesn't have major impact on tumor assessments using RECIST criteria since only the longest diameter is used.

3. Quantification

Once the pixels contributing to the typically irregular shaped lung nodule(s) is defined, nodule size is calculated using the pixel size and the length of tumor expressed in number of pixels. According to RECIST, tumor size is the longest diameter. Once the pixel co-ordinates are known, they are traversed through and distance between every possible pair of pixels is calculated. Maximum distance is the longest diameter of the tumor.

CT scan is a set of transverse image slices. Image shown in Figure 3 is one such slice. The three steps described above are repeated on all the slices and the longest diameter is calculated for each slice. Maximum of all the longest diameters is the final calculated tumor size. The seed point identified by the radiologist is stored in a text file and is read into SAS. The text file contains X, Y co-ordinates of the seed point and the slice number.

The algorithm is then applied on CT image sets acquired at different time points to obtain tumor measurements at different time points. If there is more than one tumor, then sum of longest diameters of each tumor is used to derive tumor assessments. Tumor assessments are derived using RECIST criteria as discussed in next section.

EVALUATION

The region growing algorithm is evaluated by comparing the segmentation results with the radiologist annotation that is available in the public database. The algorithm was illustrated for only one lesion. Radiologist will have to select one seed point per lesion and the above steps will have to be repeated for each lesion.

OVERALL PATIENT ASSESSMENTS

Once the individual tumor measurements are computed, overall patient assessments are derived based on the RECIST criteria. RECIST stands for Response Evaluation Criteria In Solid Tumors. It governs the rules for overall patient assessments – Complete Response, Partial Response, Stable Disease, and Disease Progression during treatment. Some of the guidelines for tumor assessment using RECIST 1.1 are presented in Appendix I.

Sum of the longest diameters of each target lesion (SLD) is computed and the response is derived for each subject and relative day based on RECIST. The rules for response evaluation for target and non-target lesions are outlined in Appendix 1. It should be noted that the criteria for Partial Response is based on change from baseline and the criteria for Progressive Disease is based on change from NADIR, which is defined as the smallest SLD since start of treatment.

Response confirmation is needed for studies where response is the primary endpoint. Confirmation of Complete Response (CR) or Partial Response (PR) is needed to deem either one of them the "best overall response" [3]. CR or PR can be claimed only if the criteria for each are confirmed at a subsequent assessment generally done 4 weeks or later. Presence of tumor assessments within 4 weeks of CR or PR presents a challenge in implementing response confirmation in SAS.

Details on SAS implementation of overall patient assessment can be found in [4], [5] and [6].

CONCLUSION AND SUMMARY

A simple segmentation algorithm was successfully implemented in SAS using PROC IML. The algorithm was evaluated using the radiologist's annotations present in the public database. This paper provides the proof of concept for implementing that end-to-end tumor assessment algorithm in SAS. Hopefully, this inspires implementation of more sophisticated tumor quantification algorithms using SAS.

REFERENCES

-
- [1] Lee W. Goldman, Principles of CT and CT Technology, J. Nucl. Med. Technol. September 2007 vol. 35 no. 3 pages 115-128
 - [2] Mathworks Forums: <http://www.mathworks.com/matlabcentral/fileexchange/19084-region-growing>
 - [3]Eisenhauer EA, Threasse P , Boagerts J et al. New response evaluation criteria in solid tumours: Revised RECIST guideline (version 1.1), EUROPEAN JOURNAL OF CANCER 45 (2009) 228 – 247
 - [4] Jian Yu, Objective tumor response and RECIST criteria in cancer clinical trials, Proceedings of 2011 Mid West SAS Users Group Conference
 - [5] Sebastien Jolivet, Main challenges for a SAS programmer stepping in SAS developer's shoes, Proceedings of PHUSE 2009
 - [6] Mei Dey, Lisa Pyle. Applying ADaM Principles in Developing a Response Analysis Dataset. Proceedings of Pharmaceutical SAS Users Group Conference 2010

ACKNOWLEDGEMENTS

We would like to thank Abhinay Joshi, Senior Imaging Scientist, of Avid Radiopharmaceuticals Inc and Aniket Joshi, Sr Research Imaging Scientist of Merck & Co Inc for their valuable insights on image processing. We would also like to thank Gopal Rajagopal, Section chair for Applications & Development, PharmaSUG 2012 for his valuable inputs.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at

Author Name: Suhas R. Sanjee
Designation : Programming Analyst
Company : Accenture LLP
Address : 2001 Market St
City: Philadelphia, State: PA Zip: 60601
Email: suhas_sanjee@merck.com

Author Name: Rajavel Ganesan
Designation : Programming Analyst
Company : Accenture LLP
Address : 2001 Market St
City: Philadelphia, State: PA Zip: 60601
Email: rajavel_ganesan@merck.com

Author Name: Mary N. Varughese
Designation : Associate Director
Company : Accenture LLP
Address : 2001 Market St
City: Philadelphia, State: PA Zip: 60601
Email: mary_varughese@merck.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

APPENDIX I: RECIST GUIDELINES

Target Lesions and Non-Target Lesions:

All lesions up to a maximum of five lesions total (and a maximum of two lesions per organ) representative of all involved organs are identified as target lesions

Target lesions should be selected on the basis of their size (lesions with the longest diameter) and their suitability for accurate repeated measurements (either by imaging techniques or clinically).

A sum of the longest diameter (LD) for all target lesions will be calculated and reported as the baseline sum LD. The baseline sum LD will be used as reference to characterize the objective tumor response.

All other lesions (or sites of disease) should be identified as non-target lesions and should also be recorded at baseline. Measurements of these lesions are not required, but the presence or absence of each should be noted throughout follow-up.

Measurable (target) lesions will be evaluated according to the following criteria

Complete response (CR): The disappearance of all target lesions with no evidence of tumor elsewhere

Partial response (PR): At least a 30% reduction in the total tumor measurement

Progressive disease (PD): Target Lesions: > 20% increase in the SLD taking as reference the smallest SLD recorded since the treatment started (nadir) and minimum 5 mm increase over the nadir. Appearance of new lesions will also lead to Progressive Disease

Stable disease (SD): Neither sufficient shrinkage to qualify for a PR nor sufficient growth to qualify for PD

Non-measurable (non-target) lesions and other lesions not identified as target lesions will be considered non-target lesions and evaluated according to RECIST, as follows:

Complete response (CR): The complete disappearance of all non-target lesions

Incomplete response (IR)/SD: The persistence of 1 or more non-target lesions

Final overall response is calculated based on target, non-target and new lesion assessments based on the following table.

Table 1: Evaluation of Overall Response by RECIST

Target Lesions	Non-Target Lesions	New Lesions	Overall Response
CR	CR	No	CR
CR	IR/SD	No	PR
PR	Non-PD	No	PR
SD	Non-PD	No	SD
PD	Any	Any	PD
Any	PD	Any	PD
Any	Any	Yes	PD